


1620 GENERAL PROGRAM LIBRARY

IBM 1620-IBM 1311 Fortran II-D Subroutine Pre-Read
and Column Test IBM Monitor I

1.2.014

**COMPUTER
TECHNOLOGY**

DR. JOHN MANIQUET
COMPUTER TECHNOLOGY DEPT.
PURDUE UNIVERSITY
CALLUMET CAMPUS
HAMMOND IN 46324



0

0

0

IBM 1620-IBM 1311 Fortran II-D Subroutine
Pre-Read and Column Test IBM 1620 Monitor I

Deck Key

- (1) SPS II-D Source Deck, 55 Cards
- (2) Fortran II-D Source Test Deck, 24 Cards

Author: Frank V. Mozina

Date: May 14, 1964

Direct inquires to: Frank V. Mozina
IBM Corporation
IBM Building
Gateway Center
Pittsburgh, Pennsylvania 15222

Modifications or revisions to this program, as they occur, will be announced in the appropriate Catalog of Programs for IBM Data Processing Systems. When such an announcement occurs, users should order a complete new program from the Program Information Department.

Program Brief

Table of Contents

Title Page	1
Deck Key	2
Table of Contents	3
Program Brief	4
Detailed Program Description	5-6
Block Diagram	7
Subroutine Source Program Listing	8-9
Subroutine Assembly Listing	10-11
Sample Program Listing	12
Sample Program Results Listing	13

1. Purpose:
This subroutine permits the Fortran program to examine columns of the next card to be read by the Fortran program, prior to the actual formatting. A call to the subroutine actually reads the next card, but also modifies the normal read routine so that the next call to the READ routine will format only this card that already has been read by the subroutine.
2. Restriction:
It is quite unlikely that this subroutine will be used with subroutine sets 1 and 3, even though the necessary addresses in these two packages agree with those in the other sets. The difficulty lies in the fact that the short form routines are overlayed, depending upon the program requirements, making it impractical to write a program in such a way as to insure that the input routines are in core at the time the FRSTRD routine is executed, and remain in core until the READ is executed.
3. Machine Configuration:
Same as Monitor I card system.
4. Program Requirements:
284 core positions
5. Source Language:
SPS II-D
6. Check-out Status:
At least 20 runs in several programs in two installations.
7. Sample problem running time:
2 minutes
8. Comments:
This program and its documentation were written by an IBM employee. It was developed for a specific purpose and submitted for general distribution to interested parties in the hope that it might prove helpful to other members of the data processing community. The program and its documentation are essentially in the author's original form. IBM serves only as the distribution agency in supplying this program. Questions concerning the use of the program should be directed to the author's attention.

Detailed Program Description

1. Subroutine Procedure:

It frequently is desirable, when dealing with mixed-format input cards, to be able to examine the next card to be read in order to determine the proper format prior to reading.

This subroutine accomplished this by reading the card into the normal input area and passing the two-digit alphabetic value for the desired column to a variable in the calling program. As many columns as desired can be examined prior to completion of the normal read.

The normal read routines are modified by the FRSTRD subroutine so that when the normal READ is executed, the input area will not be blanked, and a card will not be read. The normal READ routines are restored prior to formatting this card.

2. Incorporating the Subroutine:

Load deck number 1 as a regular monitor job. All necessary control cards are included.

Note: Refer to section entitled "Program Modification" if names other than FRSTRD and COLUMN, or subroutine numbers other than 26 and 27 are desired, prior to loading.

3. Detailed Coding Information:

The three distinct phases of this subroutine are entered as follows, and have the function described:

- a. FRSTRD is entered on the call to FRSTRD, reads the card, modifies the normal I/O, and moves the first column called for to FAC.
- b. COLUMN moves subsequent columns to FAC after initial read.
- c. RETURN is reached from the modified I/O routines and restores these routines before going to the normal formatting.

4. Program Modification:

- a. To load this program under another name, change the * NAME card to change the FRSTRD name, and the * DFLIB card to change the COLUMN name (see the source deck listing).
- b. To load the subroutine under other numbers, change the * ID NUMBER card and the * DFLIB card (see the source deck listing).

5. Operating Instructions:

In order to examine several columns of the next card to be read by the Fortran program, the user would write the following Fortran steps:

```
I1 = FRSTRD (K1)
I2 = COLUMN (K2)
I3 = COLUMN (K3)
IN = COLUMN (KN)
```

Where the I's are integer variables that are to receive the double-digit representation of the desired columns, and the K's are integer variables or constants whose values are between 1 and 80, and represent the column to be moved into the I variable.


Any amount of calculation may follow, or be interspersed with, this sequence. The next READ statement executed will read in and format the card that has been analyzed in this sequence.

Important Note: Since the output routine uses part of the same routine used by the input, do not attempt to output a record before reading the one that has been examined by a FRSTRD. The output will be in error, and the input routines will not be restored, and the card that was analyzed will be lost.

6. Sample Problem:

Deck 2 is the sample problem. After loading the subroutine, run the sample problem as a normal Monitor run. No special sense switch settings required.

Output should agree with the sample output listed in this report.



Subroutine Source Program Listing

```
ZZJOB
ZZSPS
*STORE RELOADABLE
*LIBR
*ID NUMBER0026
*NAMECOLUMN
*ASSEMBLE RELOCATABLE
*PUNCH SYMBOL TABLE
*LIST CARD
  SUB   DSA COLUMN,FRSTRD
        DORGSUB-4
        DS   5
  IORT  DS   ,565
  IOGT  DS   ,566
  DATINHDS ,06057
  PAT   DS   ,05683,ADDRESS OF OP CODE,TR
  FAC   DS   ,02492 ,PSEUDO ACCUM
  INH   DS   ,06063 ,ADDR OF INPUT AREA
  RADOITDS ,05754 ,ADDR OF READ INSTR
  STZERODS ,06239 ,ADDR OF 80 BLANKS
  *ENTRY FOR ALL AFTER FIRST COLUMN
  COLUMNTF FAC,COLUMN-1,111
        B   GCOL,,0
  *ENTRY TO PHYSICALLY READ CARD
  FRSTRDTF FAC,FRSTRD-1,111
        TR  INH-1,STZERO+1    ,,ZERO INPUT AREA
        TR  INH+86,STZERO
        TFM DATINH+2,10,10,SETFOR CARD I/O
        TFM IORT,**+23,1 , GET A CARD
        B   IOGT,DATINH-4,7
        TFM PAT,41,10, NO-OP INPUT AREA BLANKING
        TFM PAT+12,41,10, INSTRUCTIONS
        TFM RADOIT+6,RETURN,17, MAKE THE READ A BRANCH
        TFM RADOIT+1,49,10, TO THIS SUBROUTINE
  GCOL  TFM ADDR,INH-2,07, CREATE ADDRESS OF
        A   ADDR,FAC,0 , DESIRED COLUMN=
        A   ADDR,FAC,0 , 2* COL NO+INH-2
        S   FAC,FAC    ,, ZERO FAC
        A   FAC,**    ,, PUT COL VALUE IN FAC
  ADDR  DS   ,*
        BV  **+12,,0    , TURN OFF OFLOW
```

-8-

Subroutine Source Program Listing

```
BB
DORG*-9
* SUBROUTINES MODIFIED TO BRANCH HERE INSTEAD OF READ
RETURNTFM RADOIT+6,IORT,, RESTORE REGULAR
        TFM RADOIT+1,16,10, READ ROUTINE
        TFM PAT,31,10    , RESTORE ZEROING
        TFM PAT+12,31,10 , ROUTINE
        B   RADOIT+24,,, NORMAL I/O
DEND2

ZZZZ
ZZJOB
ZZDUP
*DFLIBFRSTRD 27
ZZZZ
```

-9-

Subroutine Assembly Listing

SYMBOL TABLE

STZERO 06239	RETURN 00224R	RADOIT 05754	FRSTRD 00030R	DATINH 06057
COLUMN 00006R	ADDR 00209R	FAC 02492	GCOL 00130R	INH 06063
IOGT 00566	IORT 00565	PAT 05683	SUB 00004R	

SUB	DSA	COLUMN,FRSTRD	00004	00005	-0006
			00009	00005	-0030
	DORG	SUB-4	00000		
	DS	5	00004	00005	
IORT	DS	,565	00565	00000	
IOGT	DS	,566	00566	00000	
DATINH	DS	,06057	06057	00000	
PAT	DS	,05683,ADDRESS OF OP CODE,TR	05683	00000	
FAC	DS	,02492 ,PSEUDO ACCUM	02492	00000	
INH	DS	,06063 ,ADDR OF INPUT AREA	06063	00000	
RADOIT	DS	,05754 ,ADDR OF READ INSTR	05754	00000	
STZERO	DS	,06239 ,ADDR OF 80 BLANKS	06239	00000	
*ENTRY FOR ALL AFTER FIRST COLUMN					
COLUMN	TF	FAC,COLUMN-1,111	00006	20	02492 0000N
	B	GCOL,,0	00018	M9	00150 00000
*ENTRY TO PHYSICALLY READ CARD					
FRSTRD	TF	FAC,FRSTRD-1,111	00030	20	02492 0002R
	TR	INH-1,STZERO+1 ,.ZERO INPUT AREA	00042	31	06062 06240
	TR	INH+86,STZERO	00054	31	06149 06239
	TFM	DATINH+2,10,10,SETFOR CARD I/O	00066	16	06059 000J0
	TFM	IORT,**+23,1 , GET A CARD	00078	10	00565 00101
	B	IOGT,DATINH-4,7	00090	49	00566 -6053
	TFM	PAT,41,10, NO-OP INPUT AREA BLANKING	00102	16	05683 000M1
	TFM	PAT+12,41,10, INSTRUCTIONS	00114	16	05695 000M1
	TFM	RADOIT+6,RETURN,17, MAKE THE READ A BRANCH	00126	10	05760 -0224
	TFM	RADOIT+1,49,10, TO THIS SUBROUTINE	00138	16	05755 000M9
GCOL	TFM	ADDR,INH-2,07, CREATE ADDRESS OF	00150	J6	00209 -6061
	A	ADDR,FAC,0 , DESIRED COLUMN=	00162	K1	00209 02492
	A	ADDR,FAC,0 , 2* COL NO+INH-2	00174	K1	00209 02492
	S	FAC,FAC ,. ZERO FAC	00186	22	02492 02492
	A	FAC,*-* ,. PUT COL VALUE IN FAC	00198	21	02492 00000
ADDR	DS	,*	00209	00000	

-10-

Subroutine Assembly Listing

BV	**+12,,0	, TURN OFF OFLOW	00210	M6	00222 01400
BB			00222	42	00000 00000
DORG	*-9		00224		
* SUBROUTINES MODIFIED TO BRANCH HERE INSTEAD OF READ					
RETURN	TFM	RADOIT+6,IORT,, RESTORE REGULAR	00224	16	05760 -0565
	TFM	RADOIT+1,16,10, READ ROUTINE	00236	16	05755 000J6
	TFM	PAT,31,10 , RESTORE ZEROING	00248	16	05683 000L1
	TFM	PAT+12,31,10 , ROUTINE	00260	16	05695 000L1
	B	RADOIT+24,,, NORMAL I/O	00272	49	05778 00000
DEND	2		00002		

-11-

Sample Program Listing

```
ZZJOB
ZZFORX
*PSTSN4
C   READ A CARD AND STORE COL.2 IN J
    5 J=FRSTRD(2)
C   STORE COL.5 IN K
    6 K=COLUMN(5)
C   STORE COL.10 IN L
    7 L=COLUMN(10)
C   READ THIS CARD NORMALLY
    8 READ10,M,A,B
    10 FORMAT(I2,F3.0,F5.0)
    PUNCH10,M,A,B
C   PUNCH THE VALUES FOUND IN COLS 2,5,10
    PUNCH 12,J,K,L
    12 FORMAT(3I6)
C   DO A NORMAL READ JUST TO MAKE SURE ROUTINES RESTORED
    11 READ10,M,A,B
    PUNCH10,M,A,B
    CALL EXIT
    END
    1 2 3
    4 5 6
ZZZZ
```

Sample Program Results Listing

```
00004 00002Z
00009 00005Z
00014 00010Z
00019      JZ      Z
00024      KZ      Z
00029      LZ      Z
00034      MZ      Z
00044      AZ      Z
00054      BZ      Z
0005 00056
0006 00080
0007 00104
0008 00128
0010 00194
0012 00360
0011 00392
  1 2. 3.
    71 72 73
  4 5. 6.
```